<commit_before>dnl  Copyright (C) 2010 <NAME>
dnl This file is free software; <NAME>
dnl gives unlimited permission to copy and/or distribute it,
dnl with or without modifications, as long as this notice is preserved.
 
AC_DEFUN([_PANDORA_SEARCH_THRIFT],[
  AC_REQUIRE([AC_LIB_PREFIX])
 
  dnl --------------------------------------------------------------------
  dnl  Check for thrift
  dnl --------------------------------------------------------------------
 
  AC_ARG_ENABLE([thrift],
    [AS_HELP_STRING([--disable-thrift],
      [Build with thrift support @<:@default=on@:>@])],
    [ac_enable_thrift="$enableval"],
    [ac_enable_thrift="yes"])
 
  AS_IF([test "x$ac_enable_thrift" = "xyes"],[
    AC_LANG_PUSH(C++)
    AC_LIB_HAVE_LINKFLAGS(thrift,,[
      #include <thrift/Thrift.h>
    ],[
      apache::thrift::TOutput test_output;
    ])
    AC_LANG_POP()
  ],[
    ac_cv_thrift="no"
  ])
  
  AM_CONDITIONAL(HAVE_THRIFT, [test "x${ac_cv_thrift}" = "xyes"])
  
])
 
AC_DEFUN([PANDORA_HAVE_THRIFT],[
  AC_REQUIRE([_PANDORA_SEARCH_THRIFT])
])
 
AC_DEFUN([PANDORA_REQUIRE_THRIFT],[
  AC_REQUIRE([PANDORA_HAVE_THRIFT])
  AS_IF([test x$ac_cv_thrift= xno],[
      AC_MSG_ERROR([thrift required for ${PACKAGE}])
  ])
])<commit_msg>Fix broken detection of libthrift<commit_after>dnl  Copyright (C) 2010 <NAME>
dnl This file is free software; <NAME>
dnl gives unlimited permission to copy and/or distribute it,
dnl with or without modifications, as long as this notice is preserved.
 
AC_DEFUN([_PANDORA_SEARCH_THRIFT],[
  AC_REQUIRE([AC_LIB_PREFIX])
 
  dnl --------------------------------------------------------------------
  dnl  Check for thrift
  dnl --------------------------------------------------------------------
 
  AC_ARG_ENABLE([thrift],
    [AS_HELP_STRING([--disable-thrift],
      [Build with thrift support @<:@default=on@:>@])],
    [ac_enable_thrift="$enableval"],
    [ac_enable_thrift="yes"])
 
  AS_IF([test "x$ac_enable_thrift" = "xyes"],[
    AC_LANG_PUSH(C++)
    AC_LIB_HAVE_LINKFLAGS(thrift,,[
      #include <thrift/Thrift.h>
    ],[
      apache::thrift::TOutput test_output;
    ])
    AC_LANG_POP()
  ],[
    ac_cv_libthrift="no"
  ])
  
  AM_CONDITIONAL(HAVE_LIBTHRIFT, [test "x${ac_cv_libthrift}" = "xyes"])
  
])
 
AC_DEFUN([PANDORA_HAVE_THRIFT],[
  AC_REQUIRE([_PANDORA_SEARCH_THRIFT])
])
 
AC_DEFUN([PANDORA_REQUIRE_THRIFT],[
  AC_REQUIRE([PANDORA_HAVE_THRIFT])
  AS_IF([test "x$ac_cv_libthrift" = "xno"],[
      AC_MSG_ERROR([thrift required for ${PACKAGE}])
  ])
])
{'fraction_non_alphanumeric': 0.23655913978494625, 'fraction_numerical': 0.0031859816806053365, 'mean_word_length': 2.7050147492625367, 'pattern_counts': {'":': 0, '<': 11, '<?xml version=': 0, '>': 11, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '4118870', 'n_tokens_mistral': 933, 'n_tokens_neox': 918, 'n_words': 156}